

Chocolate Games

How High School Students Discovered New Formulas Using Mathematica

Ryohei Miyadera, Shunsuke Nakamura, Yu Okada, Ryo Hanafusa, and Tomoki Ishikawa

This article presents the results of high school mathematics research projects. The authors are Dr. Miyadera and four high school students. Under the leadership of Dr. Miyadera, students have been doing mathematical research using *Mathematica* for more than 15 years, and they have discovered many formulas and theorems. They have given talks at conferences and published their results in mathematics magazines. In this article the authors present the results of the research on chocolate games, which are variants of the game of Nim. That high school students can discover new theorems and formulas with *Mathematica* is very important for mathematics education. The authors think that the best way to learn how to be creative is to make something new. The combination of *Mathematica* and the fresh minds of high school students can produce remarkable results.

■ Introduction

We study well-known chocolate games and new ones that we have invented. Most of the games presented here have simple formulas for the set of losing states, but many do not, and it is interesting that those have very beautiful graphs for the set of losing states.

■ Bitter Chocolate Games That Are Variants of Nim

The setup for Nim is a set of heaps that contain objects. For example, three heaps might contain three, four, and five objects. Two players alternate in taking any positive number of objects away from a single heap. The player who takes the last object loses.

Chocolate games generalize Nim to two dimensions. In this section we study well-known chocolate games and variations created by the authors. The chocolate games are simple to play, but they are interesting mathematically.

Definition 1

In a rectangular piece of chocolate made of squares, the brown parts are sweet and a single blue square is very bitter. Two players play alternately. To move, a player breaks the chocolate in a straight line along one of the grooves and eats the rectangular piece broken off. The player who takes the bitter square loses. The same game can be played with a rectangular box made of cubes.

There are other types of chocolate games; one of the most well-known is Chomp [1]. Again the chocolate is a rectangle made of squares, with the top-left square being very bitter. To move, a player removes a square and all the squares below it and to its right. Many people have studied this game and many interesting theories about it have been developed.

Each chocolate game treated in this article satisfies an inequality, so they are very different from Chomp, and for certain types of inequalities the mathematical structure of the winning strategy is very simple.

□ The Rectangular Chocolate Games

■ Can You Win This 2D Game?

Play the chocolate game in Figure 1 with “nim-sum” unchecked. Choose the values of x , y , z and then click “new game.” Each time you click a line (not a square), you break the chocolate into two pieces along that line and eat the piece without the bitter square [2]. The game coordinates x , y , z are the number of squares to the left, above, and to the right of the blue square, respectively.

```
Manipulate[
Module[{ab},
Column[{
Column[{
If[{x, y, z} == {0, 0, 0},
Text[
Style[
Row[{If[Mod[game1count, 2] == 0, "second", "first"],
" player won"}], Medium]],
Text[
```

```

Style[
  Row[{If[Mod[gamelcount, 2] == 0, "first", "second"],
    " player's turn"}], Medium]],
EventHandler[
  Framed@Graphics[gamelgraphics[x, y, z],
    PlotRange → {{-8, 9}, {-1, 9}},
    ImageSize → 1.2 {350, 200}, Background → Black],
  If[(Mod[gamelcount, 2] == 0 && firstplayer == "human") ||
    (Mod[gamelcount, 2] == 1 &&
      secondplayer == "human")) &&
    ! (x == 0 && y == 0 && z == 0),
  "MouseDown" :> (ab = MousePosition["Graphics"];
    {x, y, z, gamelcount} =
      cutplayer[ab, x, y, z, gamelcount])],
],
Button["click for computer to move",
  If[
    ((Mod[gamelcount, 2] == 0 &&
      firstplayer == "computer") ||
      (Mod[gamelcount, 2] == 1 &&
        secondplayer == "computer")) &&
      ! (x == 0 && y == 0 && z == 0),
    {x, y, z, gamelcount} =
      cutcomputer[x, y, z, gamelcount]],
  BaseStyle → {"GenericButton",
    If[{x, y, z} ≠ {0, 0, 0} &&
      ((Mod[gamelcount, 2] == 0 && firstplayer ==
        "computer") ||
        (Mod[gamelcount, 2] == 1 && secondplayer ==
          "computer"))}, Bold, {Plain, Gray}]],
  ImageSize → {210, 30}]
], Center],
Text@If[nimsum, Identity, Invisible]@
Column[
  {
    Style[Row[{x, "⊕", y, "⊕", z, " = ",
      BitXor[x, y, z]}], Medium],
    Style[If[BitXor[x, y, z] == 0, "losing state",
      "winning state"]],
    Grid[
      Prepend[Transpose[{
        {HoldForm[22], HoldForm[21], HoldForm[20]},
        IntegerDigits[x, 2, 3],
        IntegerDigits[y, 2, 3],
        IntegerDigits[z, 2, 3],
        IntegerDigits[BitXor[x, y, z], 2, 3]
      }]],
      {"", Style["x", Italic], Style["y", Italic],
        Style["z", Italic], "nim-sum"}], Frame → All]
  ], Alignment → Center
]

```

```

    }, Alignment → Center]
],

(* controls *)
Column[{
  Control@{{nimsum, False, "nim-sum"}, {False, True}},
  "",
  Style["dimensions relative to bitter square", Bold],
  "",
  Grid[{
    {Row[{Text@Style["x", Italic, 10], " left"}],
      Control@{{x, 7, ""}, 0, 7, 1, ImageSize → Tiny,
        Appearance → "Labeled"}},
    {Row[{Text@Style["y", Italic, 10], " above"}],
      Control@{{y, 7, ""}, 0, 7, 1, ImageSize → Tiny,
        Appearance → "Labeled"}},
    {Row[{Text@Style["z", Italic, 10], " right"}],
      Control@{{z, 7, ""}, 0, 7, 1, ImageSize → Tiny,
        Appearance → "Labeled"}}
  ]],
  "",
  Style["choice of players", Bold],
  "",
  Grid[{
    {"first", Control@{{firstplayer, "human", ""},
      {"human", "computer"}}},
    {"second",
      Control@{{secondplayer, "human", ""},
        {"human", "computer"}}}
  ]],
  "",
  "",
  Button["new game", gamelcount = 0,
    BaseStyle → {"GenericButton", 10}, ImageSize → {90, 28}],
  "",
  Style["Click a line to make a move.", Bold]
}, Alignment → Center],

ControlPlacement → Left,

Initialization := {
  gamelcount = 0;

  gamelgraphics[x_, y_, z_] :=
  {
    {Brown, Polygon[{{-x, 0}, {-x, 1+y}, {1+z, 1+y},
      {1+z, 0}}]},
    {Blue, Polygon[{{0, 0}, {0, 1}, {1, 1}, {1, 0}}]},
    Table[{Black, Line[{{-k, 0}, {-k, 1+y}}]}, {k, 0, x}],
    Table[{Black, Line[{{-x, k}, {1+z, k}}]},
      {k, 0, y+1}],
    Table[{Black, Line[{{1+k, 0}, {1+k, 1+y}}]},
      {k, 0, z}]
  }

```


```

};

cutplayer[ab_, x_, y_, z_, count_] :=
Block[{a, b, X, Y, Z, g},
{a, b} = ab;
X = Y = Z = 0;
g = 0.2;
If[
-x ≤ a ≤ 1 + z && 0 ≤ b ≤ 1 + y,
Do[If[-n - g < a < -n + g, X = x - n], {n, 0, x - 1}];
Do[If[1 + n - g < b < 1 + n + g, Y = y - n], {n, 0, y - 1}];
Do[If[1 + n - g < a < 1 + n + g, Z = z - n], {n, 0, z}];
];
If[X + Y + Z ≠ 0 && (X + Y == 0 || Y + Z == 0 || Z + X == 0),
{x - X, y - Y, z - Z, count + 1},
{x, y, z, count}
]
];

cutcomputer[x_, y_, z_, count_] :=
Block[{xyzc, randomway},
xyzc = {x, y, z, count};
randomway = RandomInteger[{0, x + y + z - 1}];
If[BitXor[x, y, z] ≠ 0,
{
If[BitXor[x, y] < z,
xyzc = {x, y, BitXor[x, y], count + 1}],
If[BitXor[y, z] < x,
xyzc = {BitXor[y, z], y, z, count + 1}],
If[BitXor[z, x] < y,
xyzc = {x, BitXor[z, x], z, count + 1}]
},
{
If[randomway ≤ x - 1,
xyzc = {randomway, y, z, count + 1}],
If[x ≤ randomway ≤ x + y - 1,
xyzc = {x, randomway - x, z, count + 1}],
If[x + y ≤ randomway,
xyzc = {x, y, randomway - x - y, count + 1}]
}
];
xyzc
]
}
]

```



▲ **Figure 1.** Chocolate game 1: a rectangular chocolate game.

In chocolate games there are two kinds of states.

Definition 2

A winning state is a state from which we can force a win as long as we play correctly at every stage. A losing state is a state from which we lose no matter how well we play, if our opponent plays correctly.

The mathematical structure of chocolate game 1 is well-known, and there is a formula to calculate losing states.

Proposition 1

The state $\{x, y, z\}$ is a losing state of chocolate game 1 if and only if $x \oplus y \oplus z = 0$, where \oplus is the bitwise exclusive-or operation (the same as `BitXor` in Mathematica).

For a proof, see [2].

Once we know the formula for losing states, the strategy to win is clear. Suppose that you have a winning state. Then you can choose a move that gives your opponent a losing state. Any move by your opponent is a winning state for you, and you can always move to a losing state again. Finally, you reach $\{0, 0, 0\}$ and win the game.

You can use this strategy in chocolate game 1 by selecting “nim-sum.” For each state $\{x, y, z\}$ you see $x \oplus y \oplus z$, and you know if the state $\{x, y, z\}$ is a losing state or a winning state.

■ Can You Win This 3D Game?

In the box chocolate game, the bitter (blue) cube is at the bottom of the 3D chocolate; to see it, drag to rotate.

Chocolate game 2 is a little more complicated than chocolate game 1, but the mathematical structure is almost the same.

In chocolate game 2, there are five controls. (You cannot cut the chocolate by clicking a face.)

Proposition 2

The state $\{v, w, x, y, z\}$ is a losing state of chocolate game 2 if and only if $v \oplus w \oplus x \oplus y \oplus z = 0$.


For a proof, see [2].

```
Manipulate[
  Graphics3D[{
    Table[{Opacity[.5], Brown, Cuboid[{a, b, c}]},
      {a, -w, x}, {b, -v, y}, {c, 0, z}], Blue,
    Cuboid[{0, 0, 0}]],
    Axes → True, PlotRange → {{-5, 6}, {-5, 6}, {0, 6}},
    ImageSize → 2 {250, 180}
  ],
  Row[{
    Column[{
      Style["cut", Bold],
      Grid[{
        {"", Control@{{y, 2, Text@Style["y", Italic, 10]}},
          0, 5, 1, Setter}, {""},
        {
          Control@{{w, 5, Text@Style["w", Italic, 10]}},
            0, 5, 1, Setter},
          Control@{{z, 5, Text@Style["z", Italic, 10]}},
            0, 5, 1, Setter},
          Control@{{x, 5, Text@Style["x", Italic, 10]}},
            0, 5, 1, Setter}
        },
        {"", Control@{{v, 4, Text@Style["v", Italic, 10]}},
          0, 5, 1, Setter}, {""}
      ]}
    ], Alignment → Center],
  Graphics3D[{
    Arrow[{{0, .5, .5}, {-5, .5, .5}}],
    Text[Style["w", Italic, 18],
      Mean[{{0, .5, .5}, {-5, .5, .5}}], {0, 1}],
    Arrow[{{.5, 0, .5}, {.5, -5, .5}}],
    Text[Style["v", Italic, 18],
      Mean[{{.5, 0, .5}, {.5, -5, .5}}], {-2, 0}],
    Arrow[{{1, .5, .5}, {5, .5, .5}}],
    Text[Style["x", Italic, 18],
      Mean[{{1, .5, .5}, {5, .5, .5}}], {0, 1}],
    Arrow[{{.5, 1, .5}, {.5, 5, .5}}],
```

```

Text[Style["y", Italic, 18],
  Mean[{{.5, 1, .5}, {.5, 5, .5}}], {-2.5, 0}],
Arrow[{{.5, .5, 1}, {.5, .5, 5}}],
Text[Style["z", Italic, 18], {.5, .5, 5}, {-1.5, 0}],
Blue, Cuboid[{0, 0, 0}
], Boxed → False, ViewAngle → 15 Degree,
PlotRange → {{-5, 6}, {-5, 5}, {0, 5}}, Axes → False,
Ticks → False, ImageSize → .65 {250, 150},
ImagePadding → .1 {{10, 10}, {10, 10}},
Background → GrayLevel[.959]
}]
]

```



▲ **Figure 2.** Chocolate game 2: a chocolate game in a box. The blue cube is underneath; to see it, drag to rotate.

□ Can You Win These New Chocolate Games?

Here are some new chocolate games.

■ *Chocolate Games That Satisfy the Inequality $2y \leq z$*

In chocolate game 3, you can cut the chocolate in three ways: to the left or right of the blue piece, or above it. The coordinates x , y , and z are the maximum number of times you can cut the chocolate in those directions.

The geometry is such that clicking lines with low values of z also deletes y lines, so that the coordinates always satisfy $2y \leq z$; that is, $y \leq \lfloor z/2 \rfloor$.

```
Manipulate[
Module[
{ab},
If[y > Floor[z / 2], y = Floor[z / 2]];

Column[
{
If[{x, y, z} == {0, 0, 0},
Text[
Style[
Row[{If[Mod[game3count, 2] == 0, "second", "first"],
" player won"}], Medium]],
Text[
Style[
Row[{If[Mod[game3count, 2] == 0, "first", "second"],
" player's turn"}], Medium]]],
EventHandler[Framed@
Graphics[game3graphics[x, y, z, (* k = *)2],
PlotRange -> {{-8, 9}, {-1, 8}},
ImageSize -> 1.2 {350, 200}, Background -> Black
],
"MouseDown" -> (ab = MousePosition["Graphics"];
{x, y, z} = cut3[ab, x, y, z, 2])
],
Text@If[nimsum, Identity, Invisible]@
Column[{
Style[Row[{x, " ⊕ ", y, " ⊕ ", z, " = ",
BitXor[x, y, z]}], Medium],
Style[If[BitXor[x, y, z] == 0, "losing state",
"winning state"]],
Grid[
Prepend[Transpose[{
HoldForm[22], HoldForm[21], HoldForm[20]}],
IntegerDigits[x, 2, 3],
IntegerDigits[Min[y, Floor[z / 2]], 2, 3],
IntegerDigits[z, 2, 3],
```

```

IntegerDigits[BitXor[x, y, z], 2, 3]]],
{"", "x", "y", "z", "nim-sum"}], Frame → All]],
Alignment → Center]
}, Alignment → Center]
],

(* controls *)
Column[
{
Control@{{nimsum, False, "nim-sum"}, {False, True}},
"",
Style["game coordinates", Bold],
"",
Control@{{x, 7, Text@Style["x", Italic, 10]}, 0,
7, 1, ImageSize → Tiny, Appearance → "Labeled"},
Control@{{y, 2, Text@Style["y", Italic, 10]}, 0,
Floor[z / 2], 1, ImageSize → Tiny,
Appearance → "Labeled"},
Control@{{z, 7, Text@Style["z", Italic, 10]}, 0,
7, 1, ImageSize → Tiny, Appearance → "Labeled"},
"",
Button[
"new game",
game3count = 0,
BaseStyle → {"GenericButton", 10}, ImageSize → {90, 28}
],
"",
Style["Click a line to make a move.", Bold],
"",
"",
Text@
Style[Row[{"With ", Style["k", Italic],
" = 2, the coordinates\n(", Style["x", Italic],
", ", Style["y", Italic], ", ", Style["z", Italic],
") = (", Dynamic@x, ", ", Dynamic@y, ", ",
Dynamic@z, ") satisfy\n", Style["y", Italic],
" ≤ [", Style["z", Italic], " / ", Style["k", Italic],
"], (",
Dynamic@y, " ≤ ", Dynamic@Floor[z / 2], " = [",
Dynamic@z, " / ", 2, ")]}.", 11]
}, Alignment → Center
],

ControlPlacement → Left,

Initialization → (
game3count = 0;

game3graphics[x_, y_, z_, k_] :=
{
{Brown,
Polygon[{


```

```

      {-x, 0}, {-x, 0},
      {(-x+1+z)/2-Max[z-(k-1)-ky, 0]/2,
       (x+1+z)/2-Max[z-(k-1)-ky, 0]/2},
      {(-x+1+z)/2+Max[z-(k-1)-ky, 0]/2,
       (x+1+z)/2-Max[z-(k-1)-ky, 0]/2},
      {1+z, 0}
    ]],
    {Blue, Polygon[{0, 0}, {-x/2, x/2},
      {(-x+1)/2, (x+1)/2}, {1, 0}]}],
    Table[{Black, Line[{-t, 0}, {-x/2-t/2, (x-t)/2}]}],
    {t, 0, x}],
    Table[
      {Black,
        Line[{(-x+1+(k-1))/2+tk/2,
          (x+1+k-1)/2+tk/2},
          {(1+z-(x+1)/2-(k-1)/2)-tk/2,
          (x+1+k-1)/2+tk/2}]}],
      {t, 0, y-Boole[y k ≤ z]}],
    Table[
      {Black,
        Line[{1+t, 0},
          {(-x+1+t)/2+Max[(t-yk-(k-1)), 0]/2,
          (x+1+t)/2-Max[(t-yk-(k-1)), 0]/2}]}],
      {t, 0, z}]
  ];

cut3[ab_, x_, y_, z_, k_] :=
  Block[
    {xyz, a, b, X, Y, Z, g},
    xyz = {x, y, z};
    {a, b} = ab;
    X = Y = Z = 0;
    g = 0.2;
    If[
      0 ≤ b && b + a > -x && b - a < x &&
      b < (x+1+z)/2-Max[z-(k-1)-ky, 0]/2 &&
      1+z > b+a,
      Do[If[-n-g < b+a < -n+g, X = x-n], {n, 0, x-1}];
      Do[If[(x+1)/2+(k-1+nk)/2-g < b <
        (x+1)/2+(k-1+nk)/2+g, Y = y-n], {n, 0, y-1}];
      Do[If[1+n-g < b+a < 1+n+g, Z = z-n], {n, 0, z-1}];
      xyz = {x-X, Min[Floor[(z-Z)/k], y-Y], z-Z}
    ];
    If[
      X+Y+Z ≠ 0 && (X+Y == 0 || Y+Z == 0 || Z+X == 0),
      game3count++
    ];
    xyz
  ]
)
]

```



▲ **Figure 3.** Chocolate game 3: a triangular chocolate game that satisfies the inequality $y \leq \lfloor z/2 \rfloor$.

Proposition 3

For $k = 2$, $\{x, y, z\}$ is a losing state of chocolate game 3 if and only if $x \oplus y \oplus z = 0$.

In fact this proposition is valid for chocolate games with the inequality $ky \leq z$ for even k . Although Proposition 3 is very similar to Proposition 1, their proofs are very different. For proofs, see [3] and [4].

■ Chocolate Games That Satisfy the Inequality $ky \leq z$

Chocolate game 4 is the same as chocolate game 3, but you can vary k . In this game $ky \leq z$; that is, $y \leq \lfloor z/k \rfloor$ for any state.

```
Manipulate[
Module[
{ab},
If[y > Floor[z / k], y = Floor[z / k]];

Column[
{
If[{x, y, z} == {0, 0, 0},
Text[
Style[
Row[{If[Mod[game3count, 2] == 0, "second", "first"],
" player won"}], Medium]],
Text[
```

```

Style[
  Row[{If[Mod[game3count, 2] == 0, "first", "second"],
    " player's turn"}], Medium]],
EventHandler[Framed@
  Graphics[game3graphics[x, y, z, (* k = *)k],
    PlotRange → {{-8, 9}, {-1, 8}},
    ImageSize → 1.2 {350, 200}, Background → Black
  ],
  "MouseDown" ⇒ (ab = MousePosition["Graphics"];
    {x, y, z} = cut3[ab, x, y, z, k])
], Alignment → Center]
],
(* controls *)
Column[
  {Style["game coordinates", Bold],
    "",
    Control@{{x, 7, Text@Style["x", Italic, 10]}, 0,
      7, 1, ImageSize → Tiny, Appearance → "Labeled"},
    Control@{{y, 2, Text@Style["y", Italic, 10]}, 0,
      Floor[z / k], 1, ImageSize → Tiny,
      Appearance → "Labeled"},
    Control@{{z, 7, Text@Style["z", Italic, 10]}, 0,
      7, 1, ImageSize → Tiny, Appearance → "Labeled"},
    Control@{{k, 2, Text@Style["k", Italic, 10]}, 1,
      4, 1, ImageSize → Tiny, Appearance → "Labeled"},
    "",
    Button[
      "new game",
      game3count = 0,
      BaseStyle → {"GenericButton", 10}, ImageSize → {90, 28}
    ],
    "",
    Style["Click a line to make a move.", Bold],
    "",
    ""
  }, Alignment → Center
],

ControlPlacement → Left,

Initialization ⇒ (
  game3count = 0;

  game3graphics[x_, y_, z_, k_] :=
  {
    {Brown,
      Polygon[{
        {-x, 0}, {-x, 0},
        {(-x + 1 + z) / 2 - Max[z - (k - 1) - k y, 0] / 2,
          (x + 1 + z) / 2 - Max[z - (k - 1) - k y, 0] / 2},
        {(-x + 1 + z) / 2 + Max[z - (k - 1) - k y, 0] / 2,
          (x + 1 + z) / 2 - Max[z - (k - 1) - k y, 0] / 2},
        {1 + z, 0}
      ]}
  }


```

```

    ]],
  {Blue, Polygon[{0, 0}, {-x/2, x/2},
    {(-x+1)/2, (x+1)/2}, {1, 0}]}],
  Table[{Black, Line[{-t, 0}, {-x/2-t/2, (x-t)/2}]}],
    {t, 0, x}],
  Table[
    {Black,
      Line[{(-x+1+(k-1))/2+t k/2,
        (x+1+k-1)/2+t k/2},
        {(1+z-(x+1)/2-(k-1)/2)-t k/2,
        (x+1+k-1)/2+t k/2}]}],
    {t, 0, y-Boole[y k ≤ z]}],
  Table[
    {Black,
      Line[{1+t, 0},
        {(-x+1+t)/2+Max[(t-y k-(k-1)), 0]/2,
        (x+1+t)/2-Max[(t-y k-(k-1)), 0]/2}]}],
    {t, 0, z}]
  };

cut3[ab_, x_, y_, z_, k_] :=
  Block[
    {xyz, a, b, X, Y, Z, g},
    xyz = {x, y, z};
    {a, b} = ab;
    X = Y = Z = 0;
    g = 0.2;
    If[
      0 ≤ b && b + a > -x && b - a < x &&
        b < (x+1+z)/2-Max[z-(k-1)-k y, 0]/2 &&
        1+z > b+a,
      Do[If[-n-g < b+a < -n+g, X = x-n], {n, 0, x-1}];
      Do[If[(x+1)/2+(k-1+n k)/2-g < b <
        (x+1)/2+(k-1+n k)/2+g, Y = y-n], {n, 0, y-1}];
      Do[If[1+n-g < b+a < 1+n+g, Z = z-n], {n, 0, z-1}];
      xyz = {x-X, Min[Floor[(z-Z)/k], y-Y], z-Z}
    ];
    If[
      X+Y+Z ≠ 0 && (X+Y == 0 || Y+Z == 0 || Z+X == 0),
      game3count++
    ];
    xyz
  ]
)
]

```



▲ **Figure 4.** Chocolate game 4: triangular chocolate games that satisfy the inequality $y \leq \lfloor z / k \rfloor$.

The authors are the first researchers who studied chocolate games whose coordinates $\{x, y, z\}$ satisfy inequalities.

For certain types of inequalities, the mathematical structure of the set of losing states is very similar to that of the well-known games of Nim and rectangular chocolate, but for most inequalities the mathematical structure of the set of losing states is very different.

■ **Chocolate Games That Satisfy the Inequality $y \leq z$**

The authors could not find any simple formula for the losing states of triangular chocolate games whose coordinates $\{x, y, z\}$ satisfy the inequality $y \leq z$. They present a conjecture for the formulas of losing states in Section 3 of [5]. It seems that chocolate games whose coordinates satisfy $ky \leq z$ for odd k do not have simple formulas for the set of losing states, because the set of losing states has very complicated graphs. In Figure 9 you can see graphs made by the set of losing states; for even k the graph is a Sierpinski gasket and for odd k the graph is complicated. As for the graphs made by losing states, see Figure 6.

■ **Chocolate Games That Satisfy the Inequality $y \leq (x + z) / k$**

```
Manipulate[
Module[
{ab},
If[yy > Floor[(xx + zz) / k], yy = Floor[(xx + zz) / k];
Column[{
If[{xx, yy, zz} == {0, 0, 0},
Text[
Style[
Row[{If[Mod[game5count2, 2] == 0, "second", "first"],
" player won"}], Medium]],
Text[Style[Row[{Dynamic@ab}], 11]];
```

```

Text[
  Style[
    Row[{If[Mod[game5count2, 2] == 0, "first", "second"],
      " player's turn"}], Medium]],
  EventHandler[Framed[game5graphics2[xx, yy, zz, k, sst]],
    {"MouseClicked" => (ab = MousePosition["Graphics"];
      {xx, yy, zz, game5count2, sst} =
        cut52[ab, xx, yy, zz, k, game5count2])}]
  ], Alignment -> Center]
],

Column[{
  Style["game coordinates", Bold],
  "",
  Control@{{xx, 7, Text@Style["x", Italic, 10]}, 0,
    7, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@{{zz, 7, Text@Style["z", Italic, 10]}, 0,
    7, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@
    {{yy, Floor[(xx + zz) / k], Text@Style["y", Italic, 10]},
    0, Floor[(xx + zz) / k], 1, ImageSize -> Tiny,
    Appearance -> "Labeled"},
  Control@{{k, 1, Text@Style["k", Italic, 10]}, 1,
    4, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  "",
  Button[
    "new game",
    game5count2 = 0,
    BaseStyle -> {"GenericButton", 10}, ImageSize -> {90, 28}
  ],
  "",
  Style["Click a line to make a move.", Bold]
  ], Alignment -> Center
],
ControlPlacement -> Left,
Initialization -> (
  game5count2 = 0;
  game5graphics2[x_, y_, z_, k_, st_] :=
    Graphics[{
      {
        Brown, Polygon[{
          {-x - 0.5, 0}, {z + 0.5, 0},

          If[
            y == Floor[(x + z) / k],
            {0.5 + z - (x + z + 1) / 2, (x + z + 1) / 2},
            {0.5 + z - k (y + 1) / 2, (k + k y) / 2}
          ],
          If[
            y == Floor[(x + z) / k],
            {- (x + 0.5) + (x + z + 1) / 2, (x + z + 1) / 2},
            {- (x + 0.5) + k (y + 1) / 2, (k + k y) / 2}
          ]
        }
      ]
    ]
  )

```




```

},
{Blue, Polygon[{{-0.5, 0}, {0.5, 0}, {0, 0.5}}]},
Table[
  Line[{
    {-0.5 - n, 0},
    If[
      Floor[(n + z) / k] ≤ y,
      {-0.5 - n + (n + z + 1) / 2, (n + z + 1) / 2},
      {-0.5 - n + (n + z + 1) / 2 - ((n + z + 1) - (y + 1) k) / 2,
        (n + z + 1) / 2 - ((n + z + 1) - (y + 1) k) / 2}
    ]
  ]], {n, 0, x}],
Table[
  Line[{
    {n + 0.5, 0},
    If[
      Floor[(n + x) / k] ≤ y,
      {0.5 + n - (n + x + 1) / 2, (n + x + 1) / 2},
      {0.5 + n - (n + x + 1) / 2 + ((n + x + 1) - (y + 1) k) / 2,
        (n + x + 1) / 2 - ((n + x + 1) - (y + 1) k) / 2}
    ]
  ]], {n, 0, z}],
Table[
  Line[{
    {-0.5 - x + Min[n k, x + z + 1] / 2, Min[n k, x + z + 1] / 2},
    {z + 0.5 - Min[n k, x + z + 1] / 2, Min[n k, x + z + 1] / 2}
  ]], {n, 0, y + 1}]
], PlotRange → {{-8, 8}, {0, 8}},
ImageSize → 1.2 {350, 200}, Background → Black];

cut52[ab_, x_, y_, z_, k_, count_] :=
Block[
  {f, a, b, xyzc},
  f = True;
  {a, b} = ab;
  xyzc = {x, y, z, count, ab};
  If[b ≤ 0 || b ≥ Min[(y + 1) k, x + z + 1] / 2 ||
    a + b ≥ z + (1 / k) || b - a ≥ x + (1 / 2), f = False];
  Do[If[Abs[-0.5 - n - (a - b)] ≤ 0.33 && f,
    xyzc = {n, Min[Floor[(n + z) / k], y], z, count + 1, ab};
    f = False, ], {n, 0, x - 1}];
  Do[If[Abs[0.5 + n - (a + b)] ≤ 0.33 && f,
    xyzc = {x, Min[Floor[(x + n) / k], y], n, count + 1, ab};
    f = False, ], {n, 0, z - 1}];
  Do[If[Abs[(n + 1) k / 2 - b] ≤ (0.33 k) && f,
    xyzc = {x, n, z, count + 1, ab}; f = False],
    {n, 0, y - 1}];
  xyzc
]
)
]

```



▲ **Figure 5.** Triangular chocolate games that satisfy the inequality $y \leq (x + z) / k$.

Conjecture 1

For a triangular chocolate game satisfying $y \leq (x + z) / k$, $\{x, y, z\}$ is a losing state if and only if $x \oplus y \oplus z = 0$ when $k \equiv 3 \pmod{4}$, and $\{x + 1, y, z + 1\}$ is a losing state if and only if $x \oplus y \oplus z = 0$ when $k \equiv 1 \pmod{4}$.

This conjecture is Proposition 4 when $k = 1$ and is proved for $k = 3$ in [6].

For other types of chocolate games that the authors studied, see [7], [6], [8], and [9].

■ Chocolate Games That Satisfy the Inequality $y \leq x + z$

```
Manipulate[
  (* The code is the same as for Figure 5 except for
     the nim-sum part and k=1. *)
  Module[
    {ab, k = 1},
    If[y > Floor[(x + z) / k], y = Floor[(x + z) / k]];

    Column[{
      If[{x, y, z} == {0, 0, 0},
        Text[
          Style[
            Row[{If[Mod[game5count, 2] == 0, "second", "first"],
              " player won"}], Medium]],
        Text[
          Style[
            Row[{If[Mod[game5count, 2] == 0, "first", "second"],
              " player's turn"}], Medium]]],
      EventHandler[
```

```

Framed@Graphics[game5graphics[x, y, z, k],
  PlotRange → {{-11, 11}, {-1, 11}},
  ImageSize → 1.2 {350, 200}, Background → Black]
, "MouseClicked" :> (ab = MousePosition["Graphics"];
  {x, y, z} = cut5[ab, x, y, z, 1])
,
Text@If[nimsum, Identity, Invisible]@
Column[
  {
    Style[Row[{"(", x, " + 1) ⊕ ", y, " ⊕ (" , z,
      " + 1) = ", BitXor[x, y, z]}], Medium],
    Style[If[BitXor[x, y, z] == 0, "losing state",
      "winning state"]],
    Grid[
      Prepend[Transpose[{
        {HoldForm[23], HoldForm[22], HoldForm[21],
          HoldForm[20]},
        IntegerDigits[x + 1, 2, 4],
        IntegerDigits[y, 2, 4],
        IntegerDigits[z + 1, 2, 4],
        IntegerDigits[BitXor[x + 1, y, z + 1], 2, 4]
      }]],
      {"", Row[{Style["x", Italic], " + 1"}],
        Style["y", Italic],
        Row[{Style["z", Italic], " + 1"}], "nim-sum"}],
      Frame → All]
    ], Alignment → Center
  ], Alignment → Center]
],
Column[{
  Control@{{nimsum, False, "nim-sum"}, {False, True}},
  "",
  Style["game coordinates", Bold],
  "",
  Grid[{
    {Row[{Text@Style["x", Italic, 10], " left"}],
      Control@{{x, 7, ""}, 0, 7, 1, ImageSize → Tiny,
        Appearance → "Labeled"}},
    {Row[{Text@Style["y", Italic, 10], " above"}],
      Control@{{y, 7, ""}, 0, 14, 1, ImageSize → Tiny,
        Appearance → "Labeled"}},
    {Row[{Text@Style["z", Italic, 10], " right"}],
      Control@{{z, 7, ""}, 0, 7, 1, ImageSize → Tiny,
        Appearance → "Labeled"}}
  ]],
  "",
  Button[
    "new game",

```

```

game5count = 0,
BaseStyle → {"GenericButton", 10}, ImageSize → {90, 28}
],
"",
Style["Click a line to make a move.", Bold]
}, Alignment → Center],

ControlPlacement → Left,

Initialization := (
game5count = 0;
game5graphics[x_, y_, z_, k_] :=
{
{
Brown, Polygon[{
{-x - 0.5, 0}, {z + 0.5, 0},

If[
y == Floor[(x + z) / k],
{0.5 + z - (x + z + 1) / 2, (x + z + 1) / 2},
{0.5 + z - k (y + 1) / 2, (k + k y) / 2}
],
If[
y == Floor[(x + z) / k],
{- (x + 0.5) + (x + z + 1) / 2, (x + z + 1) / 2},
{- (x + 0.5) + k (y + 1) / 2, (k + k y) / 2}
]]
},
{Blue, Polygon[{{-0.5, 0}, {0.5, 0}, {0, 0.5}}]},
Table[
Line[{
{-0.5 - n, 0},
If[
Floor[(n + z) / k] ≤ y,
{-0.5 - n + (n + z + 1) / 2, (n + z + 1) / 2},
{-0.5 - n + (n + z + 1) / 2 - ((n + z + 1) - (y + 1) k) / 2,
(n + z + 1) / 2 - ((n + z + 1) - (y + 1) k) / 2}
]
}], {n, 0, x}],
Table[
Line[{
{n + 0.5, 0},
If[
Floor[(n + x) / k] ≤ y,
{0.5 + n - (n + x + 1) / 2, (n + x + 1) / 2},
{0.5 + n - (n + x + 1) / 2 + ((n + x + 1) - (y + 1) k) / 2,
(n + x + 1) / 2 - ((n + x + 1) - (y + 1) k) / 2}
]
}], {n, 0, z}],
Table[
Line[{
{-0.5 - x + Min[n k, x + z + 1] / 2, Min[n k, x + z + 1] / 2},
{z + 0.5 - Min[n k, x + z + 1] / 2, Min[n k, x + z + 1] / 2}


```

```

    }], {n, 0, y+1}]
  };

cut5[ab_, x_, y_, z_, k_] :=
  Block[
    {f, a, b, xyzc},
    f = True;
    {a, b} = ab;
    xyzc = {x, y, z};
    If[b ≤ 0 || b ≥ Min[(y+1) k, x+z+1] / 2 ||
      a+b ≥ z+0.5 || b-a ≥ x+0.5, f = False];
    Do[If[Abs[-0.5-n-(a-b)] < 0.2 && f,
      xyzc = {n, Min[Floor[(n+z)/k], y], z}; f = False],
      {n, 0, x-1}];
    Do[If[Abs[0.5+n-(a+b)] ≤ 0.2 && f,
      xyzc = {x, Min[Floor[(x+n)/k], y], n}; f = False],
      {n, 0, z-1}];
    Do[If[Abs[(n+1) k / 2 - b] < 0.2 && f, xyzc = {x, n, z};
      f = False], {n, 0, y-1}];
    game5count++;
    xyzc
  ]
)
]

```



▲ **Figure 6.** The triangular chocolate game that satisfies the inequality $y \leq x + z$.

Proposition 4

For the triangular chocolate game that satisfies the inequality $y \leq x + z$, $\{x + 1, y, z + 1\}$ is a losing state if and only if $x \oplus y \oplus z = 0$.

For a proof, see [10].

■ **Chocolate Games That Satisfy the Inequalities $2x \leq y$ and $2y \leq z$**


```
Manipulate[
  If[x > Min[Floor[y / 2], Floor[z / 4]],
    x = Min[Floor[y / 2], Floor[z / 4]];
  If[y > Floor[z / 2], y = Floor[z / 2]];
  Framed@game7graphics[w, x, y, z],

Column[{
  "",
  Style["game coordinates", Bold],
  "",
  Control@{{w, 6, Text@Style["w", Italic, 10]}, 0,
    6, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  "",
  Control@{{x, 4, Text@Style["x", Italic, 10]}, 0,
    4, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@{{y, 8, Text@Style["y", Italic, 10]}, 0,
    8, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@{{z, 16, Text@Style["z", Italic, 10]}, 0,
    16, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  "",
  Text@
    Style[Row[{"The ", Style["x", Italic], " and ",
      Style["y", Italic],
      " coordinates are automatically\nadjusted to
      keep 2 ", Style["x", Italic], " ≤ ",
      Style["y", Italic], " and 2 ", Style["y", Italic],
      " ≤ ", Style["z", Italic], "."}], 10],
  "",
  Framed@Graphics3D[
    {
      Arrow[{{0, 0, 0}, {1, 0, 0}}],
      Arrow[{{0, 0, 0}, {0, 1, 0}}],
      Arrow[{{0, 0, 0}, {0, 0, 1}}],
      Arrow[{{0, 0, 0}, {0, 0, -.5}}],
      Text[Style["x", Italic], {1.2, 0, 0}],
      Text[Style["y", Italic], {0, 1.3, 0}],
      Text[Style["z", Italic], {0, 0, 1.2}],
      Text[Style["w", Italic], {0, 0, -.7}]
    }, Boxed -> False, ImageSize -> .7 {100, 100},
    ImagePadding -> 5, PlotRange -> {{0, 1}, {0, 1}, {-1, 1}},
    Background -> GrayLevel[.959]]
  ], Alignment -> Center],
```

```

ControlPlacement → Left,
Initialization → (
  game7graphics[w_, x_, y_, z_] := Block[
    {bb, bb2, ss, aa, bb0},
    ss = 4;
    aa = Flatten[Table[{a, b, c}, {a, 0, ss},
      {b, 0, 2 ss}, {c, 0, 4 ss}], 2];
    bb0 = Select[aa,
      2 #[[1]] ≤ #[[2]] && 2 #[[2]] ≤ #[[3]] &];
    bb02 = Table[{0, 0, -n}, {n, 1, 6}];
    bb = Select[bb0,
      #[[1]] ≤ x && #[[2]] ≤ y && #[[3]] ≤ z &];
    bb2 = Select[bb02, Abs[#[[3]]] ≤ w &];
    Graphics3D[
      Table[{Brown, Cuboid[bb2[[h]]]},
        {h, 1, Length[bb2]}],
      Table[{Brown, Cuboid[bb[[h]]]},
        {h, 1, Length[bb]}],
      {Blue, Cuboid[{0, 0, 0}]}
    ], Boxed → False,
    PlotRange → {{0, 4}, {0, 8}, {-6, 18}},
    ImageSize → 2 {100, 200}, Background → Black]
  ];
)
]

```



▲ **Figure 7.** The chocolate game that satisfies the inequalities $2x \leq y$ and $2y \leq z$.

Conjecture 2

For the chocolate game that satisfies the inequalities $2x \leq y$ and $2y \leq z$, the state $\{w, x, y, z\}$ is a losing state of this chocolate game if and only if $w \oplus x \oplus y \oplus z = 0$.

The authors have proved but not published this proposition.

■ Grundy Numbers and Chocolate Games

Here the authors present a *Mathematica* program to calculate losing states of the chocolate game whose coordinates satisfy the inequality $k y \leq z$. First define the function $\text{move}(\{x, y, z\})$ for each state $\{x, y, z\}$ of the chocolate and the function $\text{mex}(A)$, where A is a set of positive integers.

Definition 3

The function $\text{move}(\{x, y, z\})$ is the union of the four sets $\{\{u, y, z\}; u < x\}$, $\{\{x, v, z\}; v < y\}$, $\{\{x, v, w\}; w < z\}$, and $\{\{x, \min(\{y, \lfloor w/k \rfloor\}, w)\}; w < z\}$, where x, v, w are non-negative integers.

Definition 4

The function $\text{mex}(A)$ is the smallest non-negative integer that is not an element of the set A (the “minimum excluded ordinal”).

Example 1

$$\text{mex}(\{0, 2, 3, 4\}) = 1 \text{ and } \text{mex}(\{0, 1, 3, 4\}) = 2.$$

This defines the Grundy number $G(\{x, y, z\})$ recursively.

Definition 5

$$\text{Let } G(\{0, 0, 0\}) = 0 \text{ and } G(\{x, y, z\}) = \text{mex}(G(\{u, v, w\}); \{u, v, w\} \in \text{move}(\{x, y, z\})).$$

Proposition 5

The state $\{x, y, z\}$ is a losing state of the chocolate game if and only if $G(\{x, y, z\}) = 0$.

This is a well-known fact in the field of combinatorial games.

■ Restricting the Number of Rows and Columns You Can Remove

This game has two coordinates $\{x, y\}$ with the inequality $k y \leq x$; x and y are the height and width of the chocolate when we ignore the blue part.

See Figure 8; here h is the upper bound of the number of the rows and columns you can remove. Below the game is the table of Grundy numbers $\{G(\{x, y\}), x = 0, 1, \dots, y = 0, 1, \dots\}$.

```
Manipulate[
  If[k y ≥ x, y = Floor[x / k]];
  Module[{ab},
    Pane[
```



```

Column[{
  If[{x, y} == {0, 0},
    Text[
      Style[
        Row[{If[Mod[game8count, 2] == 0, "second", "first"],
          " player won"}], Medium]],
    Text[
      Style[
        Row[{If[Mod[game8count, 2] == 0, "first", "second"],
          " player's turn"}], Medium]]],
  EventHandler[Framed[game8graphics[x, y, k]],
    {"MouseClicked" => (ab = MousePosition["Graphics"];
      {x, y, game8count} = cut8[ab, x, y, h, k,
        game8count])}],
  "",
  Text@grundylist[x, y, h, k]
}, Alignment -> Center], ImageSize -> {250, 520}]
],

Column[{
  Style["game coordinates", Bold],
  "",
  Control@{{x, 5, Text@Style["x", Italic, 10]}, 0,
    10, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@{{y, 8, Text@Style["y", Italic, 10]}, 0,
    20, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  "",
  Control@{{h, 3, Text@Style["h", Italic, 10]}, 0,
    10, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  Control@{{k, 2, Text@Style["k", Italic, 10]}, 1,
    5, 1, ImageSize -> Tiny, Appearance -> "Labeled"},
  "",
  Button[
    "new game",
    game8count = 0,
    BaseStyle -> {"GenericButton", 10}, ImageSize -> {90, 28}
  ],
  "",
  Style["Click a line to make a move.", Bold]
}, Alignment -> Center]
],

ControlPlacement -> Left,

Initialization ->
(
  game8count = 0;

  game8graphics[x_, y_, k_] :=
  Graphics[{
    Table[{EdgeForm[Thin], Brown, Rectangle[{n, -m}],
      {m, 0, x}, {n, 0, Min[Floor[m/k], y]}],
    EdgeForm[Thin], Blue, Rectangle[{0, 0}]]

```

```

    }, Background → Black, PlotRange → {{-2, 12}, {-12, 2}},
    ImageSize → 1.2 {200, 200}];

cut8[ab_, x_, y_, h_, k_, count_] :=
Block[
  {f, temp, xyc},
  {a, b} = ab;
  f = True;
  xyc = {x, y, count};
  If[a ≤ 0 || b ≤ -x || a ≥ y + 1, f = False];
  Do[If[Abs[b + i] ≤ 0.2 && f, temp = {i, y, count};
    f = False], {i, Max[x - h, 0], x}];
  Do[If[Abs[a - i - 1] ≤ 0.2 && f, temp = {x, i, count};
    f = False], {i, Max[y - h, 0], y}];
  If[xyc ≠ temp, temp + {0, 0, 1}, xyc]
];

mex[L_] := Min[Complement[Range[0, Length[L]], L]];

move[{m_, n_}, h_, k_] := Union[
  Table[{i, Min[Floor[i / k], n]},
    {i, Max[0, m - h], m - 1}],
  Table[{m, j}, {j, Max[0, n - h], n - 1}]
];

Gr[pos_, h_, k_] :=
Gr[pos, h, k] = mex[Map[Gr[#, h, k] &, move[pos, h, k]]];

grundylistAux[{m_, n_}, h_, k_] :=
If[
  {m, n} == {-1, -1}, "",
  If[
    n == -1, m,
    If[
      m == -1, n,
      If[
        m ≥ k n, Gr[{m, n}, h, k],
        ""
      ]
    ]
  ]
];


grundylist[x_, y_, h_, k_] :=
Block[
  {p, q},
  p =
  Select[
    Select[
      Flatten[Outer[List, Range[0, x], Range[0, y]], 1],
      1 / k #[[1]] ≥ #[[2]] &
    ],
  Gr[#, h, k] == 0 &

```

```

];
q = Select[
  Flatten[Outer[List, Range[2, x + 2],
    Range[2 + Min[x, y]], 1],
    (#[[1]] - 2) < k (#[[2]] - 2) &
  ];
Row[{
  Style["x", Italic],
  Spacer[5],
  Column[{
    Style["y", Italic],
    Grid[
      Table[grundylistAux[{m, n}, h, k], {m, -1, x},
        {n, -1, y}], Frame → All, Background →
      {
        None, None,
        Join[
          Table[q[[s]] → GrayLevel[0.1],
            {s, Length[q]}],
          Table[
            {p[[s, 1]] + 2, p[[s, 2]] + 2} → Blue,
            {s, Length[p]}]
        ]
      }
    ], Alignment → Center]
}]
]
)
]

```



▲ **Figure 8.** Chocolate games that satisfy the inequality $ky \leq x$.

Conjecture 3

Let h, k be natural numbers such that $h < k$.

1. $G(\{t, kt\}) = h + 1$, for any natural number t .
2. If $ky < z$, then $G(\{y, z\}) = \text{mod}(z - y, h + 1)$.

■ *Mathematica* Programs for Chocolate Games

□ *Mathematica* Programs to Calculate Losing States

By Proposition 5, $\{x, y, z\}$ is a losing state if and only if the Grundy number $G(\{x, y, z\}) = 0$.

■ *Calculating Losing States with Grundy Numbers*

Here is a *Mathematica* program to calculate losing states using Grundy numbers. Here $k = 4$. This *Mathematica* program comes from [11]. The definition for `mex` is as in the code for Figure 8. The definitions for `move3` and `Gr3` are modifications of `move` and `Gr` from that code, so we change their names slightly.

```
mex[L_] := Min[Complement[Range[0, Length[L]], L]]

move3[{m_, n_, p_}, k_] :=
  Union[
    Table[{a, n, p}, {a, 0, m - 1}],
    Table[{m, b, p}, {b, 0, n - 1}],
    Table[{m, Min[c / k, n], c}, {c, 0, p - 1}]
  ]

Gr3[pos_, k_] :=
  Gr3[pos, k] = mex[Map[Gr3[#, k] &, move3[pos, k]]]
```

These are the candidate states for a particular case.

```
allstates[k_, s_] := allstates[k, s] =
  Select[
    Flatten[Table[{a, b, c}, {a, 0, s}, {b, 0, s}, {c, 0, s}],
      2],
    1 / k #[[3]] ≥ #[[2]] &]

allstates[4, 20]

Length@allstates[4, 20]

1386
```

This calculates the losing states using Grundy numbers.

```
losing = Select[allstates[4, 20], Gr3[#, 4] == 0 &]

{{0, 0, 0}, {1, 0, 1}, {2, 0, 2}, {3, 0, 3}, {4, 0, 4},
 {4, 1, 5}, {5, 0, 5}, {5, 1, 4}, {6, 0, 6}, {6, 1, 7},
 {7, 0, 7}, {7, 1, 6}, {8, 0, 8}, {8, 1, 9}, {8, 2, 10},
 {9, 0, 9}, {9, 1, 8}, {9, 2, 11}, {10, 0, 10}, {10, 1, 11},
 {10, 2, 8}, {11, 0, 11}, {11, 1, 10}, {11, 2, 9},
 {12, 0, 12}, {12, 1, 13}, {12, 2, 14}, {12, 3, 15},
 {13, 0, 13}, {13, 1, 12}, {13, 2, 15}, {13, 3, 14},
 {14, 0, 14}, {14, 1, 15}, {14, 2, 12}, {14, 3, 13},
 {15, 0, 15}, {15, 1, 14}, {15, 2, 13}, {15, 3, 12},
 {16, 0, 16}, {16, 1, 17}, {16, 2, 18}, {16, 3, 19},
 {16, 4, 20}, {17, 0, 17}, {17, 1, 16}, {17, 2, 19},
 {17, 3, 18}, {17, 5, 20}, {18, 0, 18}, {18, 1, 19},
 {18, 2, 16}, {18, 3, 17}, {19, 0, 19}, {19, 1, 18},
 {19, 2, 17}, {19, 3, 16}, {20, 0, 20}, {20, 4, 16}}
```

Indeed these are losing states.

```
BitXor@@@ losing

{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

There are no other losing states.

```
MemberQ[BitXor@@@ Complement[allstates[4, 20], losing],
0]

False
```

■ Calculating Losing States Using Grundy-Like Numbers

Here the authors define the Grundy-like numbers $G_2(\{x, y, z\})$, which are efficient to calculate.

Definition 6

For any state $\{x, y, z\}$, let $G_2(\{x, y, z\}) = (a, b, c)$, where (a, b, c) is defined by the following rules.

1. If $x = 0$, then let $a = 0$. If $y = 0$, then let $b = 0$. If $z = 0$, then let $c = 0$.
2. Suppose that $x \geq 1$ and $G_2(\{x-1, y, z\}) = (p, q, r)$.
 - a. Let $a = 0$, if $(p, q, r) = (0, 0, 0)$.
 - b. Let $a = p$, if $(p, q, r) \neq (0, 0, 0)$.
3. Suppose that $y \geq 1$ and $G_2(\{x, y-1, z\}) = (p, q, r)$.
 - a. Let $b = 1$, if $(p, q, r) = (0, 0, 0)$.
 - b. Let $b = q$, if $(p, q, r) \neq (0, 0, 0)$.
4. Suppose that $z \geq 1$ and $G_2(\{x, \min(z-1, y), z-1\}) = (p, q, r)$.
 - a. Let $c = 1$, if $(p, q, r) = (0, 0, 0)$.
 - b. Let $c = r$, if $(p, q, r) \neq (0, 0, 0)$.

Proposition 6

The state $\{x, y, z\}$ is a losing state of the chocolate game if and only if $G_2(\{x, y, z\}) = 0$.

For a proof of this proposition see Theorem 4.2 of [4].

This defines the corresponding *Mathematica* function G2.

```
G2[{x_, y_, z_}, k_] := G2[{x, y, z}, k] = Module[
  {w},
  w = Min[Floor[1 / k (z - 1)], y];
  {
    If[x > 0, If[Norm[G2[{x - 1, y, z}, k]] == 0, 1,
      G2[{x - 1, y, z}, k][[1]]], 0],
    If[y > 0, If[Norm[G2[{x, y - 1, z}, k]] == 0, 1,
      G2[{x, y - 1, z}, k][[2]]], 0],
    If[z > 0, If[Norm[G2[{x, w, z - 1}, k]] == 0, 1,
      G2[{x, w, z - 1}, k][[3]]], 0]
  }
]
```

This calculates the losing states using Grundy-like numbers.

```
losing2 = Select[allstates[4, 20], G2[#, 4] == {0, 0, 0} &]

{{0, 0, 0}, {1, 0, 1}, {2, 0, 2}, {3, 0, 3}, {4, 0, 4},
 {4, 1, 5}, {5, 0, 5}, {5, 1, 4}, {6, 0, 6}, {6, 1, 7},
 {7, 0, 7}, {7, 1, 6}, {8, 0, 8}, {8, 1, 9}, {8, 2, 10},
 {9, 0, 9}, {9, 1, 8}, {9, 2, 11}, {10, 0, 10}, {10, 1, 11},
 {10, 2, 8}, {11, 0, 11}, {11, 1, 10}, {11, 2, 9},
 {12, 0, 12}, {12, 1, 13}, {12, 2, 14}, {12, 3, 15},
 {13, 0, 13}, {13, 1, 12}, {13, 2, 15}, {13, 3, 14},
 {14, 0, 14}, {14, 1, 15}, {14, 2, 12}, {14, 3, 13},
 {15, 0, 15}, {15, 1, 14}, {15, 2, 13}, {15, 3, 12},
 {16, 0, 16}, {16, 1, 17}, {16, 2, 18}, {16, 3, 19},
 {16, 4, 20}, {17, 0, 17}, {17, 1, 16}, {17, 2, 19},
 {17, 3, 18}, {17, 5, 20}, {18, 0, 18}, {18, 1, 19},
 {18, 2, 16}, {18, 3, 17}, {19, 0, 19}, {19, 1, 18},
 {19, 2, 17}, {19, 3, 16}, {20, 0, 20}, {20, 4, 16}}
```

They are exactly all the losing states.

```
Apply[And, 0 == # & /@ BitXor @@@ losing2] &&
Not[MemberQ[BitXor @@@ Complement[allstates[4, 20], losing2],
  0]]

True
```


■ Graphs of the Set of Losing States

Here are some graphs of the set of losing states of chocolate games. (For more, see [10].)

■ *2D and 3D Graphs of the Set of Losing States of the Triangular Chocolate Game with $k \leq z$*

By Proposition 2 there is a simple formula for the set of losing states of the chocolate games satisfying $k \leq z$, where k is even. In Figure 9 the graph is a Sierpinski gasket for even k .

When k is odd, the graph is very different from a Sierpinski gasket. Since the graph is quite complicated, it seems likely that there is no simple formula for the set of losing states. As for the relation between the graphs of losing states and the Sierpinski gasket, see [7] (in Japanese).




▲ **Figure 9.** Graphs of the set of losing states of the game whose coordinates satisfy inequalities $k y \leq x$, for $k = 1, 2, 3, 4, 5$.

■ **2D and 3D Graphs of the Set of Losing States of the Triangular Chocolate Game with $y \leq (x + z) / k$**

Figure 10 shows that the graph is the Sierpinski gasket for odd k , so it seems that there may be a simple formula for the set of losing states in that case.

When $k = 1$, by Proposition 3 there is a simple formula for the set of losing states. When $k = 3$, the state $\{x, y, z\}$ is a losing state of this chocolate game if and only if $x \oplus y \oplus z = 0$ [6]. The authors are now trying to find a simple formula for the set of losing states for an arbitrary odd number k .

When k is even, the graph is very different from the Sierpinski gasket. Since the graph is quite complicated, it seems that there is no simple formula for the set of losing states.



▲ **Figure 10.** Graphs of the set of losing states of the games whose coordinates satisfy $y \leq (x + z)/k$ for $k = 1, 2, 3, 4, 5$.

■ Other Chocolate Games

We can study many kinds of inequalities. For example, the chocolate games that satisfy the inequalities $y \leq \left\lfloor \frac{z+h}{k} \right\rfloor$ or $y \leq h \left\lfloor \frac{z}{k} \right\rfloor$ are interesting, when k is even and h is a non-negative integer. The authors are studying these games.

■ High School Mathematical Research Project

□ How High School Students Discovered New Chocolate Games

Dr. Miyadera taught the rectangular chocolate games presented in [2] to his students and encouraged them to make new chocolate games. Students were very good at proposing new types of chocolate games, including the ones in Figures 3–8. These students were the first people who studied chocolate games that satisfy inequalities.

As for the triangular chocolate game applications on smart phones, see [12], [13], and [14].

The fact that high school students could discover new theorems using computer algebra systems is very important for mathematics education. There are many teachers around the

world who make students rediscover formulas in the classroom. In so-called “learning by doing,” students are supposed to rediscover formulas and theorems, while teachers know these formulas and theorems. In the authors’ high school research projects, students discover new things! The difference between discovery and rediscovery is very big, and the method the authors introduced can change mathematics education greatly.

The method used in the group is as follows.

First Dr. Ryohei Miyadera introduces some problems to the high school students, and after that students are required to create new problems by changing the conditions of the original problems. Sometimes they make a problem that is a little bit different from the original problem. At other times they propose a completely new problem. When the proposed problem seems to be very promising to Dr. Miyadera, he and his students begin to research it. They make computer programs using *Mathematica* and try to find interesting facts by calculation. Once they discover a new fact, they begin to prove it. Even if the proposed problem does not seem to be interesting enough, Dr. Miyadera encourages students to study it. It is often the case that a problem that seems to be trivial to him turns out to be a good problem. Some students have a kind of intuition that is beyond the imagination of Dr. Miyadera, who is an active mathematician himself. This method of research looks simple, but it is very effective.

In the history of this research group, more than 60 students have participated. Dr. Miyadera discovered that some students are very good at proposing new ideas while others are good at proving formulas and theorems. According to some researchers of mathematics education, there has never been a group of high school students that have been constantly discovering new formulas and theorems.

This group of students won the first prize in the Canada Wide Virtual Science Fair 2007, 2008, 2009, 2010, 2011, 2012, and 2013. They won the Imai Isao Award in 2007, which is the first prize in the Japan Wide High School Research Contest supported by Kogakuin University in Japan. They won the first prize in the Japan Science & Engineering Challenge 2008, and represented Japan in the International Science and Engineering Fair. (Japan declined to send them to the fair because of the epidemic of swine influenza.) They became finalists in the Japan Science & Engineering Challenge in 2011, 2012, and 2013. They became Google Science Fair Regional Finalists in 2012 and became semifinalists in the Yau High School Mathematics Awards in 2012 and 2013.

For a detailed explanation about the authors’ high school mathematics research project, see [15], [16], [17], [18], [19], [20], and [21]. The authors have presented their research at the International *Mathematica* Symposium in [22], [23], and [24].

They have also presented their research in the *Mathematica* Demonstrations Project, and Ryohei Miyadera was selected as a featured contributor. See [25], [26], and [27].

Dr. Miyadera received the Excellent Teacher Award from the Ministry of Education of Japan in 2012, and he received the Wolfram Innovator Award in 2012 for his research with high school students.

If you are interested in doing a high school or undergraduate mathematics research project, one of the best books to use is by Cowen and Kennedy [11].

■ Conclusion

If students use *Mathematica* freely with an experienced mathematician, they can discover a lot of new and interesting facts of mathematics. The difference between a true discovery and rediscovery is great. We think that the best way to learn how to be creative is to create new and interesting things.

■ Acknowledgments

We would like to thank Professor Robert Cowen of the City University of New York for his encouragement when we met [22]; he gave his book [11] to us. We learned how to use *Mathematica* in our research by using his book. We also would like to thank Ed Pegg Jr of Wolfram Research and Professor Tadashi Takahashi for giving us valuable advice.

■ References

- [1] Wikipedia. "Chomp." (Dec 4, 2013) en.wikipedia.org/wiki/Chomp.
- [2] A. C. Robin, "A Poisoned Chocolate Problem," Problem Corner, *The Mathematical Gazette*, **73**(466), 1989 pp. 341–343. (An answer for the above problem is in **74**(468), 1990 pp. 171–173.)
- [3] R. Miyadera, S. Nakamura, and R. Hanafusa, "New Chocolate Games—Variants of the Game of Nim," *Proceedings of the Annual International Conference on Computational Mathematics, Computational Geometry & Statistics*, Singapore, 2012 pp. 122–128.
dl4.globalstf.org/?wpsc-product=new-chocolate-game-variants-of-the-game-of-nim.
- [4] R. Miyadera, S. Nakamura, and R. Hanafusa, "New Chocolate Games," *GSTF Journal of Mathematics, Statistics and Operations Research*, **1**(1), 2012 pp. 111–116.
dl4.globalstf.org/?wpsc-product=new-chocolate-games.
- [5] S. Nakamura, Y. Naito, R. Miyadera et al., "Chocolate Games That Are Variants of Nim and Interesting Graphs Made by These Games," *Visual Mathematics*, **14**(2), 2012.
www.mi.sanu.ac.rs/vismath/miyaderasept2012/index.html.
- [6] T. Yamauchi, T. Inoue, and Y. Tomari, "Variants of the Game of Nim That Have Inequalities as Conditions," *Rose-Hulman Institute of Technology, Undergraduate Math Journal*, **10**(2), 2009. www.rose-hulman.edu/mathjournal/archives/2009/vol10-n2/paper12/v10n2-12pd.pdf.
- [7] R. Miyadera and S. Nakamura, "A Chocolate Game," *Mathematics of Puzzles, Journal of Information Processing* (special issue), **53**(6), 2012 pp. 1582–1591 (in Japanese).
- [8] M. Naito, T. Yamauchi, T. Inoue et al., "Discrete Mathematics and Computer Algebra System," *Proceedings of the 9th Asian Symposium on Computer Mathematics-Mathematical Aspects of Computer and Information Sciences Joint Conference (ASCM-MACIS)*, Fukuoka, Japan: Kyushu University, 2009 pp. 127–136.
gcoe-mi.jp/temp/publish/a2019bae287f4628e4bfc9d273150000.pdf.
- [9] MathPuzzle.com. "The Bitter Chocolate Problem." Material added 8 Jan 06 (Happy New Year). www.mathpuzzle.com/26Feb2006.html.
- [10] T. Nakaoka, R. Miyadera et al., "Combinatorial Games and Beautiful Graphs Produced by Them," *Visual Mathematics*, **11**(3), 2009.
www.mi.sanu.ac.rs/vismath/miyaderasept2009/index.html.

- [11] R. Cowen and J. Kennedy, *Discovering Mathematics with Mathematica*, Erudition Books, 2001.
 - [12] M. Fukui, “Bitter Chocolate Games 1.”
itunes.apple.com/jp/app/bitter-chocolate-games-1/id495880660?mt=8.
 - [13] M. Fukui, “Bitter Chocolate Games 2.”
itunes.apple.com/jp/app/bitter-chocolate-games-2/id502476677?mt=8.
 - [14] S. Nakamura, “Poison Chocolate Game.”
market.android.com/details?id=com.kgshmathclub.chocoapp.
 - [15] R. Miyadera, T. Inoue, and S. Nakamura, “High School Mathematics Research Project Using Computer Algebra System,” *Proceedings of the 16th International Seminar on Education of Gifted Students in Mathematics*, The Korean Society of Mathematical Education, Chungnam National University, Daejeon, Korea, 2011 pp. 303–316.
 - [16] S. Hashiba, D. Minematsu, H. Shimoda, and R. Miyadera, “How High School Students Can Discover Original Ideas of Mathematics Using *Mathematica*,” *Mathematica in Education and Research* **11**(3), 2006.
 - [17] H. Matsui, D. Minematsu, T. Yamauchi, and R. Miyadera, “Pascal-Like Triangles and Fibonacci-Like Sequences,” *Mathematical Gazette*, **94**(529), 2010 pp. 27–41.
 - [18] R. Miyadera, T. Hashiba, Y. Nakagawa, T. Yamauchi, H. Matsui, S. Hashiba, D. Minematsu, and M. Sakaguchi, “Pascal-Like Triangles and Sierpinski-Like Gaskets,” *Visual Mathematics*, **9**(1), 2007. www.mi.sanu.ac.rs/vismath/miyad/pascaliketriangle.html.
 - [19] H. Matsui and T. Yamauchi, “Formulas for Fibonacci-Like Sequences Produced by Pascal-Like Triangles,” *Rose-Hulman Institute of Technology Undergraduate Math Journal*, **9**(2), 2008. www.rose-hulman.edu/mathjournal/archives/2008/vol9-n2/paper11/v9n2-11p.pdf.
 - [20] H. Matsui, N. Saita, K. Kawata, Y. Sakurama, and R. Miyadera, “Elementary Problems, B-1019,” *Fibonacci Quarterly*, **44**(3), 2006.
www.fq.math.ca/Problems/August2006elementary.pdf.
 - [21] R. Miyadera, “General Theory of Russian Roulette.”
library.wolfram.com/infocenter/MathSource/5710.
 - [22] R. Miyadera, K. Fujii et al., “How High School Students Could Present Original Math Research Using *Mathematica*,” The International *Mathematica* Symposium, 2003.
library.wolfram.com/infocenter/Conferences/4983.
 - [23] R. Miyadera and D. Minematsu, “Curious Properties of an Iterative Process: How Could High School Students Present Original Mathematics Research Using *Mathematica*?,” The International *Mathematica* Symposium, 2004. library.wolfram.com/infocenter/Conferences/6055.
 - [24] R. Miyadera, “Using *Mathematica* in Doing the Research of Mathematics with High School Students,” *Proceedings of the Fourth International Mathematica Symposium (IMS2001)*, Tokyo, 2001.
 - [25] H. Matsui, T. Yamauchi, D. Minematsu, and R. Miyadera. “Pascal-Like Triangles Made from a Game” from the Wolfram Demonstrations Project—A Wolfram Web Resource.
demonstrations.wolfram.com/PascalLikeTrianglesMadeFromAGame.
 - [26] H. Matsui, T. Yamauchi, D. Minematsu, and R. Miyadera. “Pascal-Like Triangles Mod k ” from the Wolfram Demonstrations Project—A Wolfram Web Resource.
demonstrations.wolfram.com/PascalLikeTrianglesModK.
 - [27] T. Inoue, T. Nakaoka, and R. Miyadera. “A Bitter Chocolate Problem That Satisfies an Inequality” from the Wolfram Demonstrations Project—A Wolfram Web Resource.
demonstrations.wolfram.com/ABitterChocolateProblemThatSatisfiesAnInequality.
- R. Miyadera, S. Nakamura, Y. Okada, T. Ishikawa, and R. Hanafusa, “Chocolate Games,” *The Mathematica Journal*, 2013. [dx.doi.org/doi:10.3888/tmj.15-12](https://doi.org/10.3888/tmj.15-12).

About the Authors

Ryohei Miyadera received a Ph.D. (mathematics) at Osaka City University and received a second Ph.D. (mathematics education) at Kobe University. He has two fields of research: probability theory of functions with values in an abstract space and applications of *Mathematica* to discrete mathematics. He and his high school students have been doing research in discrete mathematics for more than 15 years. They have talked at IMS 2001 [24], IMS 2003 [22], and IMS 2004 [23]. They have also published more than 22 refereed articles. Ryohei Miyadera received the Excellent Teacher Award from the Ministry of Education of Japan in 2012, and he received the Wolfram Innovator Award in 2012. His hobby is long-distance running, and he is one of the best runners in the over-55-years-old group of his prefecture.

Shunsuke Nakamura has graduated from Kwansei Gakuin High School and is preparing for the entrance examination to university. He did mathematical research with Dr. Miyadera when he was a high school student at Kwansei Gakuin. He has published papers in [3], [4], [5], and [7]. He is a member of the Information Processing Society of Japan. Nakamura's hobbies are playing the piano and playing video games.

Yu Okada is a high school student. He is a very serious player of Massively Multiplayer Online RPGs.

Tomoki Ishikawa is a university student at Kwansei Gakuin. He did mathematical research with Dr. Miyadera when he was a high school student at Kwansei Gakuin. He has published papers on origami mathematics. He is a photographer who received an excellent photography award in the Japan Wide Cultural Festival.

Ryo Hanafusa is a university student at Kwansei Gakuin. He did mathematical research with Dr. Miyadera when he was a high school student at Kwansei Gakuin. He has published papers in [3] and [4]. He is a photographer who received an excellent photography award in the Japan Wide Cultural Festival.

Ryohei Miyadera

*Mathematics Department
Kwansei Gakuin High School
runners@kwansei.ac.jp
miyadera1272000@yahoo.co.jp*

Shunsuke Nakamura

*Kwansei Gakuin High School
nakashun1994@gmail.com*

Yu Okada

*Kwansei Gakuin High School
alm1_a1m_aim@ezweb.ne.jp*

Tomoki Ishikawa

*Kwansei Gakuin High School
tom94826@gmail.com*

Ryo Hanafusa

*Kwansei Gakuin High School
ryo3waygate@yahoo.co.jp*